Select Statement to generate result, for group by per second and link with Table
How about creating a table holding all times for a day and using it to drive the query?

In this simple demo, I set up a table similar to yours, load it with data, then use the dbo.timeLookup table to drive it.  The example assumes you are only working with one day, but it could be extended if your timespan needed to span multiple days:

USE tempdb
GO

SET NOCOUNT ON

-- Create a large table
IF OBJECT_ID('dbo.user_table') IS NOT NULL DROP TABLE dbo.user_table
CREATE TABLE dbo.user_table ( 
    rowId INT IDENTITY PRIMARY KEY, 
    someData UNIQUEIDENTIFIER DEFAULT NEWID(), 
    session_datetime DATETIME DEFAULT GETDATE(), 
    addedBy VARCHAR(30) DEFAULT SUSER_NAME(), 
    ts ROWVERSION
    )
GO


IF OBJECT_ID('dbo.timeLookup') IS NOT NULL DROP TABLE dbo.timeLookup
CREATE TABLE dbo.timeLookup ( 
    [time]      TIME ,
    [hour]      AS DATEPART( hour, [time] ),
    [minute]    AS DATEPART( minute, [time] ),
    [second]    AS DATEPART( second, [time] ),

    CONSTRAINT PK_dbo_timeLookup PRIMARY KEY ( [time] ),
    CONSTRAINT UK_dbo_timeLookup UNIQUE ( [hour], [minute], [second] )
    )
GO


-- Load up some dummy data
;WITH cte AS (
SELECT TOP 1000000 ROW_NUMBER() OVER ( ORDER BY ( SELECT 1 ) ) rn
FROM master.sys.columns c1
    CROSS JOIN master.sys.columns c2
    CROSS JOIN master.sys.columns c3
)
INSERT INTO dbo.user_table ( session_datetime )
SELECT DATEADD( second, rn % 86400, '09/29/2014' )
FROM cte
WHERE rn % 33 != 0  
GO

-- Suggested index
CREATE INDEX _idx ON dbo.user_table ( session_datetime )
GO

-- Populate the timeLookup table
;WITH cte AS (
SELECT TOP 86400 ROW_NUMBER() OVER ( ORDER BY ( SELECT 1 ) ) rn
FROM master.sys.columns c1
    CROSS JOIN master.sys.columns c2
    CROSS JOIN master.sys.columns c3
)
INSERT INTO dbo.timeLookup ( [time] )
SELECT DATEADD( second, rn, '1 Jan 2014' )
FROM cte
GO

-- Remove certain seconds for the demo
DELETE dbo.user_table
WHERE  CAST( session_datetime AS TIME ) In ( '15:10:08', '15:10:09' )

-- Have a look at the sample records loaded
--SELECT MIN(session_datetime) xmin, MAX(session_datetime) xmax, COUNT(*) records
--FROM [User_Table]


-- Main query
SELECT tl.[time], COUNT(ut.session_datetime) AS records
FROM dbo.timeLookup tl
    LEFT JOIN dbo.user_table ut ON tl.[time] = CAST( ut.session_datetime AS TIME )
        AND ut.[session_datetime] Between '2014-09-29 00:00:00.000' And '2014-09-29 23:59:59.999'
WHERE tl.[time] Between '15:10:06' And '15:10:10'
GROUP BY tl.[time]


Result: